Rules for integrands of the form $(c\,x)^m P_q[x] \left(a\,x^j + b\,x^n\right)^p$

1: $\int P_q[x^n] \left(a\,x^j + b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land -1 < n < 1$

Derivation: Integration by substitution

Basis: If $d \in \mathbb{Z}^+$, then $F[x^n] = d\,\text{Subst}\left[x^{d-1} F\left[x^{d\,n}\right], x, x^{1/d}\right] \partial_x x^{1/d}$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land -1 < n < 1$, let $d = \text{Denominator}[n]$, then

$$\int P_q[x^n] \left(a\,x^j + b\,x^n\right)^p dx \longrightarrow d\,\text{Subst}\left[\int x^{d-1} P_q\left[x^{d\,n}\right] \left(a\,x^{d\,j} + b\,x^{d\,n}\right)^p dx,\ x,\ x^{1/d}\right]$$

Program code:

```
Int[Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  With[{d=Denominator[n]},
  d*Subst[Int[x^(d-1)*ReplaceAll[SubstFor[x^n,Pq,x],x→x^(d*n)]*(a*x^(d*j)+b*x^(d*n))^p,x],x,x^(1/d)]] /;
FreeQ[{a,b,j,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && RationalQ[j,n] && IntegerQ[j/n] && LtQ[-1,n,1]
```

2. $\int (c\,x)^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Note: If $n\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}$, then $m\in\mathbb{Z}$, and $(c\,x)^m$ automatically evaluates to $c^m\,x^m$.

Rule: If $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx\,\longrightarrow\,\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,P_q[x]\,\left(a\,x^{j/n}+b\,x\right)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*SubstFor[x^n,Pq,x]*(a*x^Simplify[j/n]+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,j,m,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (c\,x)^m P_q[x^n] \left(a\,x^j + b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z}$, then

$$\int (c\,x)^m P_q[x^n] \left(a\,x^j + b\,x^n\right)^p dx \longrightarrow \frac{(c\,x)^m}{x^m} \int x^m P_q[x^n] \left(a\,x^j + b\,x^n\right)^p dx$$

Program code:

```
Int[(c_*x_)^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  c^(Sign[m]*Quotient[m,Sign[m]])*(c*x)^Mod[m,Sign[m]]/x^Mod[m,Sign[m]]*Int[x^m*Pq*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] &&
  IntegerQ[Simplify[(m+1)/n]] && RationalQ[m] && GtQ[m^2,1]
```

```
Int[(c_*x_)^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^m/x^m*Int[x^m*Pq*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[(m+1)/n]]
```

3. $\int (c\,x)^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,(j\mid n)\in\mathbb{Z}^+$

1: $\int x^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,\left(j\mid n\mid\frac{i}{n}\right)\in\mathbb{Z}^+\land\,m\in\mathbb{Z}\,\land\,\text{GCD}[m+1,\,n]\neq 1$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}$, let $g=\text{GCD}[m+1,\,n]$, then $x^m\,F[x^n]=\frac{1}{g}\,\text{Subst}\left[x^{\frac{m+1}{g}-1}\,F\left[x^{\frac{n}{g}}\right],\,x,\,x^g\right]\,\partial_x x^g$

Rule: If $p\notin\mathbb{Z}\,\land\,\left(j\mid n\mid\frac{i}{n}\right)\in\mathbb{Z}^+\land\,m\in\mathbb{Z}$, let $g=\text{GCD}[m+1,\,n]$, if $g\neq 1$, then

$$\int x^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx\;\longrightarrow\;\frac{1}{g}\,\text{Subst}\left[\int x^{\frac{m+1}{g}-1}\,P_q\left[x^{\frac{n}{g}}\right]\,\left(a\,x^{\frac{j}{g}}+b\,x^{\frac{n}{g}}\right)^p\,dx,\,x,\,x^g\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  With[{g=GCD[m+1,n]},
  1/g*Subst[Int[x^((m+1)/g-1)*ReplaceAll[Pq,x→x^(1/g)]*(a*x^(j/g)+b*x^(n/g))^p,x],x,x^g] /;
 NeQ[g,1]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && IGtQ[j,0] && IGtQ[n,0] && IGtQ[j/n,0] && IntegerQ[m]
```

2: $\int (c\,x)^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,(j\mid n)\in\mathbb{Z}^+\land\,j<n\,\land\,q>n-1\,\land\,m+q+n\,p+1\neq 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule: If $p\notin\mathbb{Z}\,\land\,(j\mid n)\in\mathbb{Z}^+\land\,j<n\,\land\,q>n-1\,\land\,m+q+n\,p+1\neq 0$, then

$$\int (c\,x)^m\,P_q[x^n]\,\left(a\,x^j+b\,x^n\right)^p\,dx\,\longrightarrow$$

$$\int (c\,x)^m\,\left(P_q[x^n]-P_q[x,q]\,x^q\right)\,\left(a\,x^j+b\,x^n\right)^p\,dx+\frac{P_q[x,q]}{c^q}\int (c\,x)^{m+q}\,\left(a\,x^j+b\,x^n\right)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,q]\,(c\,x)^{m+q-n+1}\,\left(a\,x^j+b\,x^n\right)^{p+1}}{b\,c^{q-n+1}\,(m+q+n\,p+1)}+$$
$$\int (c\,x)^m\,\left(P_q[x^n]-P_q[x,q]\,x^q-\frac{a\,P_q[x,q]\,(m+q-n+1)\,x^{q-n}}{b\,(m+q+n\,p+1)}\right)\,\left(a\,x^j+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    Pqq*(c*x)^(m+q-n+1)*(a*x^j+b*x^n)^(p+1)/(b*c^(q-n+1)*(m+q+n*p+1)) +
    Int[(c*x)^m*ExpandToSum[Pq-Pqq*x^q-a*Pqq*(m+q-n+1)*x^(q-n)/(b*(m+q+n*p+1)),x]*(a*x^j+b*x^n)^p,x]] /;
  GtQ[q,n-1] && NeQ[m+q+n*p+1,0] && (IntegerQ[2*p] || IntegerQ[p+(q+1)/(2*n)])] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && Not[IntegerQ[p]] && IGtQ[j,0] && IGtQ[n,0] && LtQ[j,n]
```

4. $\int (c\,x)^m P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx$ when $p\notin\mathbb{Z}\land j\neq n\land \frac{j}{n}\in\mathbb{Z}\land \frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx$ when $p\notin\mathbb{Z}\land j\neq n\land \frac{j}{n}\in\mathbb{Z}\land \frac{n}{m+1}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m F[x^n] = \frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right], x, x^{m+1}\right]\partial_x x^{m+1}$

Rule: If $p\notin\mathbb{Z}\land j\neq n\land \frac{j}{n}\in\mathbb{Z}\land \frac{n}{m+1}\in\mathbb{Z}$

$$\int x^m P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx \longrightarrow \frac{1}{m+1}\,\text{Subst}\left[\int P_q\left[x^{\frac{n}{m+1}}\right]\left(a\,x^{\frac{j}{m+1}}+b\,x^{\frac{n}{m+1}}\right)^p dx,\ x,\ x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  1/(m+1)*Subst[
    Int[ReplaceAll[SubstFor[x^n,Pq,x],x→x^Simplify[n/(m+1)]]*(a*x^Simplify[j/(m+1)]+b*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,j,m,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] &&
  IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (c\,x)^m\, P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} = 0$

Rule: If $p\notin\mathbb{Z}\,\land\,j\neq n\,\land\,\frac{j}{n}\in\mathbb{Z}\,\land\,\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (c\,x)^m\, P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx \longrightarrow \frac{(c\,x)^m}{x^m}\int x^m\, P_q[x^n]\,(a\,x^j+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  c^(Sign[m]*Quotient[m,Sign[m]])*(c*x)^Mod[m,Sign[m]]/x^Mod[m,Sign[m]]*Int[x^m*Pq*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] &&
  IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]] && GtQ[m^2,1]
```

```
Int[(c_*x_)^m_*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^m/x^m*Int[x^m*Pq*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && PolyQ[Pq,x^n] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] &&
  IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

5: $\int (c\,x)^m\,P_q[x]\,\left(a\,x^j+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}\land j\neq n$

Derivation: Algebraic expansion

Rule:

$$\int (c\,x)^m\,P_q[x]\,\left(a\,x^j+b\,x^n\right)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(c\,x)^m\,P_q[x]\,\left(a\,x^j+b\,x^n\right)^p,\,x\right]dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*Pq*(a*x^j+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,j,m,n,p},x] && (PolyQ[Pq,x] || PolyQ[Pq,x^n]) && Not[IntegerQ[p]] && NeQ[n,j]
```

```
Int[Pq_*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a*x^j+b*x^n)^p,x],x] /;
FreeQ[{a,b,j,n,p},x] && (PolyQ[Pq,x] || PolyQ[Pq,x^n]) && Not[IntegerQ[p]] && NeQ[n,j]
```